

COMPUTERS & TYPESETTING



DONALD E. KNUTH

COMPUTERS & TYPESETTING / E



Computer Modern Typefaces

```

cmchar "Lowercase Greek chi";
beginchar(oct "037", 11u#, x_height#, desc_depth#);
italcorr x_height# * slant + .5hair# - 1.5u#;
adjust_fit(0, 0); pickup fine.nib;
pos1(hair, 180); pos2(vair, 80); pos5(vair, 80); pos6(hair, 180);
x1 - .5hair = hround -.5hair; x6 = w - x1; x2 = w - x5 = u;
y1 = y2l - h/8; top y2r = h + oo; bot y5l = -d - oo; y6 = y5r + h/8;
z0 = .5[z2, z5]; z2' = z2l + (.5u, 0); z5' = z5r - (.5u, 0);
numeric theta; theta = angle(z5' - z2') + 90;
pos3(stem, theta); pos4(stem, theta);
y3r = y2r - h/8; y4l = y5l + h/8; z3l = whatever[z2', z0]; z4r = whatever[z0, z5'];
filldraw stroke z1e{up} ... z2e{right} ... z3e{z4r - z3l}
    .. z4e{z4r - z3l} ... {right}z5e ... {up}z6e; % major diagonal and hooks
x8 = w - x7 = 2u; y7 + .5hair = h; y8 - .5hair = -d;
numeric theta; theta = angle(z7 - z8) - 90;
pos7(hair, theta); pos8(hair, theta);
filldraw z7l --- z8l .. z8r --- z7r .. cycle; % minor diagonal
math_fit((desc_depth# - x_height#) * slant + ic#, ic#);
penlabels(0, 1, 2, 2', 3, 4, 5', 5, 6, 7, 8); endchar;

cmchar "Lowercase Greek psi";
beginchar(oct "040", 11u#, asc_height#, desc_depth#);
italcorr x_height# * slant;
adjust_fit(0, 0); pickup fine.nib; interim superness := more_super;
x0 = 0; x2 = x3 + .25u; pos3(stem, -180); lft x3r = hround(2.5u - .5stem);
hook_in(0, 1, 2)(skewed); % opening hook
pos2'(stem, -180); z2' = z2; pos4(vair, -90); pos5(hair, 0);
x4 = .5w + u; rt x5r = hround(w - .5u);
y3 = .7[y4, y5]; bot y4r = -oo; y5 = .57x_height;
filldraw stroke z2'e{-u, -x_height} ... super_arce(3, 4)
    .. tension atleast 1.05 .. {up}z5e; % left stem and arc
v_bulb(5, 6); % closing bulb
pos7(hair, 0); pos8(hair, 0); x7 = x8 = .5w + .5u; top y7 = h; bot y8 = -d - oo;
filldraw stroke z7e -- z8e; % stem
math_fit(-2/3x_height# * slant + .5hair# + .5u#, 2/3ic#);
penlabels(0, 1, 2, 3, 4, 5, 6, 7, 8); endchar;

```





```

cmchar "Lowercase Greek omega";
beginchar(oct "041", 11u#, x_height#, 0);
italcorr x_height# * slant;
adjust_fit(0, 0); pickup fine.nib;
pos2(hair, -180); lft x2r = hround .5u; y2 = y4 = y6 = y8 = bar_height;
y1 + .5hair = h; x1 = x2 + .75u; pos1(hair + dw, angle(2(x1 - x2), y1 - y2) + 90);
pos3(vstem + dw, -90); pos4(hair, 0); pos5(vair, 90);
pos6(hair, 180); pos7(vstem + dw, 270); pos8(hair, 360);
bot y3r = bot y7r = -oo; top y5r = vround(.2[y4, h] + .5vair);
x3 = .5[x2, x4]; x7 - .25u = .5[x6, x8]; rt x8r = hround(w - .5u);
x5 + .5stem = hround(.5w + .5stem) + eps;
x5 = .5[x4, x6]; rt x4r - lft x6r = min(stem, 2hair) + 2eps;
if x4l < x6l: x4l := x6l := x5; fi
filldraw circ_stroke z1e{2(x2 - x1), y2 - y1} ... z2e{down} ... z3e{right}
... {up} z4e ... {left} z5e; % left arc
filldraw stroke z5e{left} ... z6e{down} ... z7e{right} ... {up} z8e; % right arc
v_bulb(8, 9); % closing bulb
math_fit(-bar_height# * slant, 2/3ic#); penlabels(1, 2, 3, 4, 5, 6, 7, 8, 9); endchar;

cmchar "Variant lowercase Greek epsilon";
beginchar(oct "042", 8u#, x_height#, 0);
italcorr x_height# * slant - .5u#;
adjust_fit(0, 0); pickup fine.nib;
numeric light_vstem; light_vstem = Vround 2/3[vair, vstem];
pos2(light_vstem, 90); x2 = x8 = .5w + .5u; top y2r = h + o;
x1 + .5light_vstem = hround(w - u); y1 = .2[y2, bar_height];
pos1(light_vstem, angle(x1 - x2, 2(y1 - y2)) + 90); pos3(hair, 180); pos4(vair, 270);
pos5(hair, 360); pos6(vair, 450); pos7(hair, 540); pos8(light_vstem, 630);
lft x3r = hround u; x4 = x6 = .5w; rt x5r = hround(w - 2.5u); x7 = x3;
y3 = .5[y2, y4]; y5 = bar_height = .5[y4, y6]; y7 = .5[y6, y8]; bot y8r = -o;
top y6r - bot y4r = vstem + eps; if y6l < y4l: y6l := y4l := y5; fi
x9 + .5vair = hround(w - .5u); y9 = 1/3[y8, bar_height];
pos9(vair, angle(x8 - x9, 2(y8 - y9)) + 90);
filldraw circ_stroke z1e{x2 - x1, 2(y2 - y1)} ... pulled_super_arce(2, 3)(.5superpull)
& super_arce(3, 4) ... {up} z5e; % upper arc
filldraw circ_stroke z9e{x8 - x9, 8(y8 - y9)} ... pulled_arce(8, 7)
& super_arce(7, 6) ... {down} z5e; % lower arc
math_fit(-.5bar_height# * slant - .5u#, ic#);
penlabels(1, 2, 3, 4, 5, 6, 7, 8, 9); endchar;

```



```

cmchar "Variant lowercase Greek theta";
beginchar(oct "043", 9u#, asc_height#, 0);
italcorr .5[bar_height#, x_height#] * slant + .5vair# + .5u#;
adjust_fit(0, 0); pickup fine.nib;
numeric light_curve; light_curve = hround .5[stem, curve];
x0 = 0; x2 - .5stem = hround(2.5u - .5stem); hook_in(0, 1, 2); % opening hook
pos2'(stem, -180); z2' = z2; pos3(stem, -180); pos4(vair, -90);
pos5(light_curve, 0); pos6(vair, 90); pos7(hair, 180); pos8(vair, 270);
x3 = x2; x4 = x6 = .5[x3, x5]; rt x5r = hround(w - 1.5u + .5light_curve);
lft x7r = hround(3.25u - .5hair); x8 + .5vair = hround(w + .5vair - epsilon);
y3 = .4x_height; bot y4r = -oo; y5 = .5[y4, y6]; top y6r = h + oo;
y7 = .5[x_height, h]; y8 = .5[bar_height, x_height];
filldraw stroke z2'e .. {{ interim superness := hein_super; super_arc_e(3, 4) }}
    & pulled_arc_e(4, 5) & pulled_arc_e(5, 6)
    ... {down} z7e ... {4(x8 - x7), y8 - y7} z8e; % bowl and loop
math_fit(-2/3x_height# * slant + .5hair# + .5u#, ic#);
penlabels(0, a, 1, 2, 3, 4, 5, 6, 7, 8); endchar;

cmchar "Variant lowercase Greek pi";
beginchar(oct "044", 14.5u#, x_height#, 0);
italcorr x_height# * slant + .5stem# - u#;
adjust_fit(0, 0); pi_stroke; % hook and bar
pos4(hair, -180); pos5(hair, -180); pos6(vstem, -90);
pos7(hair, 0); pos7'(hair, -180); z7' = z7;
pos8(vstem, -90); pos9(hair, 0); pos0(hair, 0);
x4 = x5 + 1.5u; lft x5r = hround 2u;
x6 = x4 + .5u; lft x7l = hround(.5w + .25u - .5hair);
x8 = w - 4.5u; rt x9r = hround(w - 1.5u); x0 = x9 - 1.25u;
top y4 = top y0 = h; y5 + .1h = y7 - .1h = y9 = bar_height; bot y6r = bot y8r = -oo;
filldraw stroke z4e {2(x5 - x4), y5 - y4} ... z5e {down}
    ... z6e {right} ... z7e {up}; % left bowl
filldraw stroke z0e {2(x9 - x0), y9 - y0} ... z9e {down}
    ... z8e {left} ... z7'e {up}; % right bowl
math_fit(-.7x_height# * slant + .5hair# + .5u#, ic# - .5u#);
penlabels(0, 1, 2, 3, 4, 5, 6, 7, 8, 9); endchar;

```

